SINGULAR VALUE FACTORIZATION

PURPOSE

Compute the singular value factorization of a matrix.

DESCRIPTION

If X is a matrix with row and column dimensions n and p respectively, then an n by n orthogonal matrix U and a p by p orthogonal matrix V can be found such that:

$$U^T X V = \begin{bmatrix} \Sigma \\ 0 \end{bmatrix} \qquad \textbf{(EQ 4-76)}$$

where Σ is a m by m diagonal matrix (m is the minimum of n and p). The diagonal elements of Σ are the singular values of X and they are stored from largest to smallest. The above assumes that n >= p. A right hand side becomes $[\Sigma\ 0]$ if N < p. Singular values of zero (or near zero) indicate that the matrix is singular (i.e., not of full rank) or ill-conditioned. Chapters 2 and 14 of the Numerical Recipes book describe some applications of the SVD.

Since U and V are orhogonal (and so their inverses are equal to their transpose), the above equation can also be written as:

$$X = U \begin{bmatrix} \Sigma \\ 0 \end{bmatrix} V^T \qquad \textbf{(EQ 4-77)}$$

For large matrices, it can be impractical to compute U (which is n by n). However, U can be partitioned into

U = (U1, U2)

where U1 is n by p. Then

$X = U1\Sigma V'$

is called the singular value factorization of X. Several multivariate statistical techniques are based on this factorization. The program example demonstrates the biplot proposed by Ruben Gabriel.

SYNTAX

LET <u> <s> <v> = SINGULAR VALUE FACTORIZATION <mat> <SUBSET/EXCEPT/FOR qualification>
where <mat> is a matrix for which the singular values are to be computed;
<u> is an n by p matrix where U is saved;
<s> is a variable where the singular values are saved (length is minimum of n and p);
<v> is an p by p matrix where V is saved.
and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

EXAMPLES

LET U S V = SINGULAR VALUE DECOMPOSITION A

NOTE 1

DATAPLOT uses the LINPACK routine SSVDC to calculate the singular value factorization.

NOTE 2

DATAPLOT will calculate the singular value decomposition even if N <= p. However, in practice this is almost never done.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

MATRIX EIGENVALUES	=	Compute the matrix eigenvalues.
MATRIX EIGENVECTORS	=	Compute the matrix eigenvectors.

MATRIX MULTIPLICATION	=	Perform a matrix multiplication.
MATRIX SOLUTION	=	Solve a system of linear equations.
CORRELATION MATRIX	=	Compute the correlation matrix of a matrix.
VARIANCE-COVA MATRIX	=	Compute the variance-covariance matrix of a matrix.
SINGULAR VALUES	=	Compute the singular values of a matrix.
SINGULAR VALUE DECOM	=	Compute the singular value decomposition of a matrix.

REFERENCE

"LINPACK User's Guide," Dongarra, Bunch, Moler, Stewart. Siam, 1979.

"Numerical Recipes: The Art of Scientific Programming (FORTRAN Version)," Press, Flannery, Teukolsky, and Vetterling, Cambridge University Press, 1989 (chapter 2).

APPLICATIONS

Linear Algebra, Multivariate Analysis

IMPLEMENTATION DATE

93/8

PROGRAM

```
. Generate a biplot (derived from the singular value factorization)
. SOURCE: "THE BIPLOT AS A DIAGNOSTIC TOOL FOR MODELS OF TWO-WAY
. TABLES", BRANDU, GABRIEL, TECHNOMETRICS, FEB. 1978.
. DATA IS YIELDS OF COTTON, ROWS ARE VARIETY, COLUMNS ARE CENTER
DIMENSION 100 COLUMNS
READ MATRIX X
 1.55 1.26 1.41 1.78
 3.39 3.47 2.82 3.89
 1.95 1.91 1.74 2.29
10.47 9.12 9.55 17.78
 1.45 1.51 1.41 1.70
 3.72 3.55 3.09 4.27
 4.47 4.07 3.98 4.47
END OF DATA
LET N = SIZE X1
.
FEEDBACK OFF
LET P = MATRIX NUMBER OF COLUMNS X
LOOP FOR K = 1 1 P
    LET X^K = LOG10(X^K)
END OF LOOP
LET SUM1 = 0
LOOP FOR K = 1 1 P
     LET TEMP = SUM X^K
    LET SUM1 = SUM1 + TEMP
END OF LOOP
LET GMEAN = SUM1/(N*P)
LET X = MATRIX SUBTRACTION X GMEAN
LET U S V = SINGULAR VALUE FACTORIZATION X
LET DENOM = MATRIX EUCLIDEAN NORM X
LET S1 = S(1)
LET S2 = S(2)
LET GF = (S1**2 + S2**2)/DENOM**2
.
LET B = MATRIX TRANSPOSE V
LET U1 = U1*SQRT(S1)
LET U2 = U2*SQRT(S2)
LET B1 = B1*SQRT(S1)
```

```
LET B2 = B2*SQRT(S2)
.
LET TAG = SEQUENCE 1 1 N
LET TAG2 = SEQUENCE 1 1 P
CHARACTER CIRCLE SQUARE
CHARACTER FILL SOLID ALL
LINE BLANK ALL
TITLE BIPLOT
X1LABEL GOODNESS OF FIT = ^GF
LEGEND FILL SOLID
LEGEND FONT SIMPLEX
LEGEND 1 SQUA() - COLUMN MARKERS
LEGEND 2 CIRC() - ROW MARKERS
.
PLOT U2 U1 AND
PLOT B2 B1
LEGEND 1
LEGEND 2
LIMITS FREEZE
PRE-ERASE OFF
CHARACTER 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20
CHARACTER OFFSET 1.2 0 ALL
PLOT U2 U1 TAG
PLOT B2 B1 TAG2
```